######################################################################
# grandOthello qmake
######################################################################

# app indica ao qmake que estamos montando uma aplicacão
TEMPLATE = app

# Diretórios onde estarão outros arquivos do projeto
DEPENDPATH += src
INCLUDEPATH += .
INCLUDEPATH += src

# Pasta onde os arquivos temporarios de compilacao ficaram armazenados
OBJECTS_DIR = tmp
MOC_DIR = tmp
RCC_DIR = tmp

# Diretorio onde os arquivos executaveis serao gerados
DESTDIR = bin

# Indica ao qmake que vamos utilizar a biblioteca qt
# com os modulos core e gui (que sao os que utilizaremos)
CONFIG += qt

# Indica se deve criar um makefile com suporte aos simbolos de debug ou nao
#CONFIG += debug

CONFIG += release
DEFINES += QT_NO_DEBUG_OUTPUT
DEFINES += QT_NO_WARNING_OUTPUT

# Habilitando os warnings do compilador no makefile.
CONFIG += warn_on

# Se estivermos em uma plataforma Windows
win32 {
    # Se estiver em modo debug, incluimos as bibliotecas de console para podermos
    # utilizar funcoes de debug como a qDebug():
    # http://doc.qtsoftware.com/4.5/qtglobal.html#qDebug
    debug {
	CONFIG += console
    }

    # Então configura de acordo.
    CONFIG += windows
}

# Se estivermos em ambiente unix
unix {
    CONFIG += x11
}



# Versao do grandOthello =]
VERSION = 1.0.1

# Nome do executavel criado
TARGET = grandOthello

## ADICIONEM OS ARQUIVOS AQUI ##
# Arquivos .h
HEADERS += src/Coordenada.h
HEADERS += src/Jogador.h
HEADERS += src/MinMax.h
HEADERS += src/Tabuleiro.h
HEADERS += src/GrandOthello.h
HEADERS += src/FuncaoAvaliacao.h
HEADERS += src/FuncaoPerimetroBonus.h
HEADERS += src/FuncaoMatrizEstatica.h
HEADERS += src/FuncaoQuantidade.h
HEADERS += src/Jogada.h
HEADERS += src/MainWindow.h
HEADERS += src/GamePanel.h
HEADERS += src/defines.h
HEADERS += src/ConfigWindow.h

# Arquivos .cpp
SOURCES += src/Coordenada.cpp
SOURCES += src/Jogador.cpp
SOURCES += src/MinMax.cpp
SOURCES += src/Tabuleiro.cpp
SOURCES += src/GrandOthello.cpp
SOURCES += src/FuncaoAvaliacao.cpp
SOURCES += src/FuncaoPerimetroBonus.cpp
SOURCES += src/FuncaoMatrizEstatica.cpp
SOURCES += src/FuncaoQuantidade.cpp
SOURCES += src/Jogada.cpp
SOURCES += src/MainWindow.cpp
SOURCES += src/GamePanel.cpp
SOURCES += src/ConfigWindow.cpp

# Arquivos .ui (QTDesigner)
#FORMS += src/meuform.ui

# Arquivos dos resources (imagens por exemplo) do jogo
RESOURCES += grandOthello.qrc
RC_FILE = imgs/grandOthello.rc
